Please write your answers to the following questions on separate sheets of paper (one page per question). Question point values are shown in parentheses. The exam is worth a total of 62 points.

1. (12) What output would be generated from the following click event?

```vbnet
Private Sub cmdQ1_Click()
    Dim i As Integer, k As Integer
    Dim sum1 As Integer, sum2 As Integer
    Dim sum3 As Integer
    For i = 1 To 3
        sum1 = sum1 + i
        For k = 1 To sum1 Step 2
            sum2 = sum2 + k
            sum3 = sum3 + 1
        Next k
    Next i
    Form1.Print sum1, sum2, sum3
End Sub
```

2. (10) In class we looked at a procedure similar to the following that reversed the contents of the cells of an array that was passed to it as an argument:

```vbnet
Public Sub reverse(a() As Integer, ByVal n As Integer)
    Dim k As Integer, temp As Integer
    Dim j As Integer
    j = n
    For k = 1 To n / 2
        temp = a(k)
        a(k) = a(j)
        a(j) = temp
        j = j - 1
    Next k
End Sub
```

Rewrite this procedure so that it uses the Do While … Loop construct instead of the For…Next construct.
3. (10) On Exam 1 you were asked to write a click event that determined if a phrase entered into a text box was a palindrome (a phrase that was spelled the same way forward and backward). The user repeatedly clicked the command button until the status of the phrase was determined.

The code below shows a solution to that problem:

```vba
Private Sub cmdCheckLetters_Click()
    Static letterPos As Integer
    Dim word As String
    letterPos = letterPos + 1
    word = txtWord.Text
    If letterPos >= (Len(word) - letterPos + 1) Then
        MsgBox "It’s a palindrome"
        letterPos = 0
        Exit Sub
    End If
    If Mid$(word, letterPos, 1) <> Mid$(word, Len(word) - letterPos + 1, 1) Then
        MsgBox "Not a palindrome"
        letterPos = 0
    End If
End Sub
```

Rewrite this click event so that it determines the status of the phrase in just one click. That is, the user enters text into txtWord and then clicks on cmdCheckLetters. In response to this click, the status “It’s a palindrome” or “Not a palindrome” is displayed.

4. (10) Consider the following statement: “A recordset is equivalent to a set of records that compose a table in a relational database.”

Is this a correct or incorrect statement? Explain your answer.
5. (20) You are given the following click event:

```vbnet
Private Sub cmdQ5_Click()
    Dim a(1 To 50) As Integer
    Dim modalValue As Integer
    Dim i As Integer
    ' generate 50 random numbers between 1 and 10
    For i = 1 To 50
        a(i) = Rnd() * 9 + 1
    Next i
    modalValue = mode(a()) ' write this function
    Form1.Print "Mode is " & modalValue
End Sub
```

Write the complete mode function that is referenced in the third from the last line above. The modal value is the value that occurs most frequently in a set of values. For example, the modal value in the following list of numbers is 4:

4, 5, 2, 4, 5, 6, 1, 6, 4

Your function should work for any one-dimensional integer array that stores values ranging from 1 to 10. The function may assume that the values stored in the elements of the array are legal (values 1 through 10). If there are two or more values that occur with the same maximum frequency, the function should return the first one. For example, consider the following list of numbers:

2, 5, 8, 6, 2, 6, 7, 1, 6, 2

Both the numbers 2 and 6 occur 3 times so in this case the mode function should return a 2.

First outline your solution approach (use some form of pseudocode to do this). Then write valid VB code to implement your solution approach.
Sample Answers:

1. The values printed for sum1, sum2, and sum3 are:
   6   14   6

2. 
   ```
   Public Sub reverse(a() As Integer, ByVal n As Integer)
   Dim k As Integer, temp As Integer
   Dim j As Integer
   k = 1
   j = n
   Do While k < j
       temp = a(k)
       a(k) = a(j)
       a(j) = temp
       k = k + 1
       j = j - 1
   Loop
   End Sub
   ```

3. 
   ```
   Private Sub cmdCheckLetters_Click()
   Dim letterPos As Integer
   Dim word As String
   word = txtWord.Text
   For letterPos = 1 To Len(word) / 2
       If Mid$(word, letterPos, 1) <> Mid$(word, Len(word) - letterPos + 1, 1) Then
           MsgBox "Not a palindrome"
           letterPos = 0
           Exit Sub
       End If
   Next letterPos
   MsgBox "It's a palindrome"
   letterPos = 0
   End Sub
   ```

4. The statement is only partially correct. A recordset can be the same as a table from a database, but it can also be associated with a set of records that are the result of an SQL query. In this case, the recordset is not associated with an actual table in the database.
5.

```vba
Public Function mode(x() As Integer) As Integer
    Dim count(1 To 10) As Integer
    Dim k As Integer
    Dim max As Integer, maxLoc As Integer
    'Find the frequency counts
    For k = 1 To UBound(x)
        count(x(k)) = count(x(k)) + 1
    Next k
    'Find the max frequency
    For k = 1 To 10
        If count(k) > max Then
            max = count(k)
            maxLoc = k
        End If
    Next k
    mode = maxLoc
End Function
```